SA06/Introduction to Maple
Math 112, Spring 2006

SA06: (1.4) 2(a), 6; (1.5) 7; and the two Maple problems at the end of this handout.

Introduction to Maple. In this homework, you will learn the basics of using Maple by doing a few computations. Instructions are given for the Windows computers in the Math Lab; if you bought your own copy of Maple, instructions should be included.

Running Maple. To run Maple, under the Windows "Start" button, go to the "Math Apps" menu, and select Maple 9.5. If there is a particular package or worksheet for you to use in this assignment, go to "My Computer", go to the "handouts" drive, double-click on the "hsu" folder, and double-click on the appropriate package or worksheet.

Basic Maple. Here are some of the basic Maple commands you'll need this term. We'll assume that you're typing the commands in at the `>` prompt, though if you prefer to use the menu buttons at the top to do the same things, that's fine.

In this course, you can think of Maple as a very sophisticated desktop calculator. For instance:

```
> 47*59*71;
```

$$196883$$

The important thing to notice here is that every Maple command must end with a semicolon. If you omit the semicolon, Maple thinks that you're typing in a very long command and haven't finished yet.

The standard math operators include the usual `+`, `-`, `*`, `/`, `^`, and `sqrt`. Most standard mathematical functions are included (e.g., `exp(x)`, Maple's version of e^x, `sin()`, `ln()`, `arctan()`, etc.), and the number π is represented *exactly* by the constant symbol `Pi`. This brings up another important point, namely, that Maple is designed to work with exact values, and not just with numerical approximations:

```
> arctan(sin(Pi/2));
```

$$\frac{1}{4}\pi$$

This can be very helpful for doing mathematical computations, but can also produce some unexpected, and even perverse, results. Using the command `simplify` can help matters. For instance:

```
> sqrt(4);
```

$$\sqrt{4}$$

```
> simplify(%);
```

$$2$$

(The symbol `%` used above means "the answer to the previous command.") On the other hand, Maple's idea of simplification doesn't always agree with yours, so sometimes you'll actually want to use `expand` instead:

```
> (x-1)*(x+1);
```

$$(x-1)(x+1)$$

```
> simplify(%);
```

$$(x-1)(x+1)$$

```
> expand(%);
```

$$x^2 - 1$$

To get numerical values, use the command `evalf()`. Beware of roundoff error, though:

```
> evalf(Pi);
```

$$3.141592654$$

```
> sin(%);
```

$$-.4102067615\ 10^{-9}$$

```
> sin(Pi);
```

$$0$$

Any name (sequence of letters and numbers, starting with a letter) which Maple doesn't know already is assumed to be an unspecified mathematical unknown. This can produce some odd-looking effects if you're not prepared, especially if you forget that all multiplication must be explicitly stated with the `*` operator:

```
> ax*x;
```

$$ax\ x$$

```
> a*x*x;
```

$$a\ x^2$$

```
> mc*mc;
```

$$mc^2$$

Note the subtle difference in the output of the last two examples. In Maple output, mc^2 is definitely not the same as $m\ c^2$.

Names can also be used as variables. Assignment is done with the `:=` operator:

```
> y := (x+3);
```

$$y := x + 3$$

```
> expand(y^3);
```

$$x^3 + 9x^2 + 27x + 27$$

One of Maple's strengths is its ability to do symbolic computations from calculus. For instance:

```
> limit(sin(3*x)/x,x=0);
```

$$3$$

```
> diff(exp(-x),x);
```

$$-\mathbf{e}^{(-x)}$$

```
> int(1/(1+z^2),z);
```

$$\arctan(z)$$

```
> int(1/(1+z^2),z=0..infinity);
```

$$\frac{1}{2}\pi$$

Remember that the only way to get e^x in Maple is to use `exp(x)`; in particular, the only way to get the constant $e = 2.71828\ldots$ is `exp(1)`. Also, even though Maple omits the $+C$ in an indefinite integral, you should still include it when appropriate.

Another of Maple's strengths is solving equations. For instance:

```
> curve := solve(y^2=x^3+x^2,y);
```

$$curve := \sqrt{x+1}\, x, -\sqrt{x+1}\, x$$

Note that we have also stored the solution in the variable `curve`. More precisely, `curve[1]` is now the name of the first solution, and `curve[2]` is the name of the second.

The main way in which we'll use Maple, however, is to make pictures. For instance, the command

```
> plot(x/2+1/2,x=-1..2);
```

draws the graph $y = \frac{x}{2} + \frac{1}{2}$ for $-1 \leq x \leq 2$. You can also draw the graph of the result of a `solve` command; try:

```
> plot({curve[1],curve[2]},x=-1..2);
```

Note that this command uses the results of our previous `solve` command. Also, be careful to use curly braces {} around `curve[1],curve[2]`, and not just parentheses.

It is also useful to know that you can combine several plots into one. First, you have to load the `plots` Maple package:

```
> with(plots):
```

Note that we finished the command with a colon and not a semicolon. This tells Maple to do the command, but to keep quiet about the results. In any case, we next save the results of our plots in variables:

```
> curveplot := plot({curve[1],curve[2]},x=-1..2):
```

```
> lineplot := plot(x/2+1/2,x=-1..2):
```

Again, note the colons. The combined plots are displayed with the command:

```
> display({curveplot,lineplot});
```

Note that we go back to using a semicolon, since we want to see the results of the computation.

Finally, if you want to print out what you've done, pull down the "File" menu and select "Print". *Warning:* this will print out your entire worksheet, that is, everything you've done, mistakes and all. If you're printing out something for an assignment, you can save lots of paper by first deleting everything that you don't want to print. (To do this, just select the unwanted sections with the mouse, and press Delete.)

Problems: For each of the following problems, turn in a printout of the computation used to solve the problem and the answer to the problem.

1. Compute the indefinite integral $\int e^{x/a} \sin(-bx)\, dx$, where a and b are arbitrary nonzero constants, and check your answer by finding its derivative with respect to x.
2. Graph the functions $\sin 2x$ and $\cos \sqrt{x}$ $(0 \le x \le 2\pi)$ on the same set of axes.

More Maple: graphs of functions of 2 variables in 3-space. Here is where Maple starts to come in very handy for this class. To plot, for instance, $z = x^2 + y^2$, do:

```
> plot3d(x^2+y^2,x=-2..2,y=-2..2,axes=normal);
```

Unlike `plot`, `plot3d` requires you to specifically request axes to appear. If you omit the `axes=normal` part of the command, none will appear.

A really cool thing: To see this 3-D plot (or any 3-D plot) from a different viewpoint, click in the picture window with the mouse and drag the picture around. Try it — I hope it will make the rest of this handout seem worthwhile. Note also that while you have a plot selected, you can change various attributes of the plot, such as the axes style, the colors, and so on.

There is also a command `display3d`, which is part of the `plots` package, and is entirely analogous to the `display` command for 2-D plots.

Help! For command-line help, use the `?` command. For instance, to get help on the command `plots3d`, do:

```
> ?plots3d;
```

Maple also has a nice menu-driven help system. In particular, for a very complete introduction to Maple and all of its features, select "New User's Tour" under the Help menu and follow the ensuing instructions.